Variation of Parameters and Duhamel's Principle

Date: Feb 26, 2002
Last Revision: Feb 26, 2002
Maple 6

Bent E. Petersen
bent@alum.mit.edu
petersen@math.orst.edu

Course: Mth 256
Term: Winter 2002
File name: 256w2002_varparm_duhamel.mws

```
> restart;
```

In this worksheet we illustrate Duhamel's principle and variation of parameters. We use Maple to ease the calculations though, of course, if one is going to use Maple there is generally little reason to do variation of parameters. Our reason here is pedagogical.

We consider only second order linear differential equations. Corresponding results hold for higher order.

Consider a linear inhomogeneous ordinary differential equation of the form

```
> ode00:=diff(y(t),t,t)+p(t)*diff(y(t),t)+q(t)*y(t)=g(t);
```

$$ode00 := \left(\frac{\partial^2}{\partial t^2} y(t)\right) + p(t)\left(\frac{\partial}{\partial t} y(t)\right) + q(t)\, y(t) = g(t)$$

The method of variation of parameters gives us a particular solution of ode00 in the form

```
> y(t)=Int(K(t,s)*g(s),s=a..t);
```

$$y(t) = \int_a^t K(t, s)\, g(s)\, ds$$

where

```
> K(t,s)=(y1(s)*y2(t)-y1(t)*y2(s))/(y1(s)*D(y2)(s)-D(y1)(s)*y2(s));
```

$$K(t, s) = \frac{y1(s)\, y2(t) - y1(t)\, y2(s)}{y1(s)\, D(y2)(s) - D(y1)(s)\, y2(s)}$$

and y1, y2 is a fundamental solution set for the associated homogeneous equation.

For each s we see K(t,s) is a linear combination of y1(t) and y2(t) and so is a solution of the associated linear homogeneous ordinary differential equation

```
> ode00h:=lhs(ode00)=0;
```

$$ode00h := \left(\frac{\partial^2}{\partial t^2} y(t)\right) + p(t)\left(\frac{\partial}{\partial t} y(t)\right) + q(t)\, y(t) = 0$$

Moreover, it is clearly the solution with initial values

```
> y(s)=0,D(y)(s)=1;
```

$$y(s) = 0,\ D(y)(s) = 1$$

This characterization of K(t,s) as the solution (for each s) of an initial value problem for a *homogeneous* linear ordinary differential equation, and the corresponding result for higher order equations, is known as Duhamel's principle. Note while K(t,s) is defined in terms of a fundamental solution set, Duhamel's principle shows us that it doesn't actually depend on the choice of fundamental solution set.

Duhamel's principle has a physical interpretation. It also has theoretical applications.

Here then is a procedure to calculate K(t,s):

```
> duhamel:=proc(ode,y,t,s)
>   local odeh,inith,solnh,K;
> odeh:=lhs(ode)=0;
> inith:=y(s)=0,D(y)(s)=1;
> K:=rhs(dsolve({odeh,inith},y(t)));
> end:
```

Note this procedure is not very robust. We make no error checking and we make strong assumptions about the way our differential equation is written.

Let's look at some examples:

Example 1

```
> ode01:=(x-1)*diff(y(x),x,x)-x*diff(y(x),x)+y(x)=(x-1)^2/x;
```

$$ode01 := (x-1)\left(\frac{\partial^2}{\partial x^2} y(x)\right) - x\left(\frac{\partial}{\partial x} y(x)\right) + y(x) = \frac{(x-1)^2}{x}$$

```
> K01:=duhamel(ode01,y,x,s);
```

$$K01 := -\frac{x}{s-1} + \frac{s\,\mathbf{e}^x}{\mathbf{e}^s\,(s-1)}$$

```
> g01:=subs(x=s,rhs(ode01)/(x-1));
```

$$g01 := \frac{s-1}{s}$$

```
> int(K01*g01,s): y01:=simplify(subs(s=x,%));
```

$$y01 := -x\ln(x) - 1$$

Let's compare our result with Maple's solution of ode01.

```
> dsolve(ode01,y(x));
```

$$\mathrm{y}(x) = -x\ln(x) - 1 + \_C1\,x + \_C2\,\mathbf{e}^x$$

Setting the parameters _C1 and _C2 to 0 we a particular solution and it is the same as the one we found by Duhamel's principle.

Example 02

```
> ode02:=diff(y(x),x,x)-y(x)=exp(2*x)/(1+exp(x));
```

$$ode02 := \left(\frac{\partial^2}{\partial x^2}\,\mathrm{y}(x)\right) - \mathrm{y}(x) = \frac{\mathbf{e}^{(2\,x)}}{1+\mathbf{e}^x}$$

```
> K02:=duhamel(ode02,y,x,s);
```

$$K02 := \frac{1}{2}\,\frac{\mathbf{e}^x}{\mathbf{e}^s} - \frac{1}{2}\,\mathbf{e}^s\,\mathbf{e}^{(-x)}$$

```
> g02:=subs(x=s,rhs(ode02));
```

$$g02 := \frac{\mathbf{e}^{(2\,s)}}{1+\mathbf{e}^s}$$

```
> int(K02*g02,s): y02:=simplify(subs(s=x,%));
```

$$y02 := \frac{1}{2}\,\mathbf{e}^x\ln(1+\mathbf{e}^x) - \frac{1}{4}\,\mathbf{e}^x + \frac{1}{2} - \frac{1}{2}\,\mathbf{e}^{(-x)}\ln(1+\mathbf{e}^x)$$

Again let's compare our result with Maple's solution of ode02.

```
> dsolve(ode02,y(x));
```

$$\mathrm{y}(x) = \frac{1}{2}\,\mathbf{e}^x\ln(1+\mathbf{e}^x) - \frac{1}{4}\,\mathbf{e}^x + \frac{1}{2} - \frac{1}{2}\,\mathbf{e}^{(-x)}\ln(1+\mathbf{e}^x) + \_C1\,\mathbf{e}^x + \_C2\,\mathbf{e}^{(-x)}$$

Example 03

```
> ode03:=diff(y(t),t,t)+y(t)=tan(t);
```

$$ode03 := \left(\frac{\partial^2}{\partial t^2} y(t)\right) + y(t) = \tan(t)$$

```
> K03:=duhamel(ode03,y,t,s);
```

$$K03 := \cos(s)\sin(t) - \sin(s)\cos(t)$$

```
> g03:=subs(t=s,rhs(ode03));
```

$$g03 := \tan(s)$$

```
> int(K03*g03,s): y03:=simplify(subs(s=t,%));
```

$$y03 := -\cos(t)\ln\left(\frac{1+\sin(t)}{\cos(t)}\right)$$

Again let's compare our result with Maple's solution of ode03.

```
> dsolve(ode03,y(t));
```

$$y(t) = -\cos(t)\ln(\sec(t)+\tan(t)) + \_C1\sin(t) + \_C2\cos(t)$$

Again we have agreement.

Example 04

```
> ode04:=x^2*diff(y(x),x,x)+x*diff(y(x),x)+(x^2-1/4)*y(x)=x^(5/2)*co
  s(x);
```

$$ode04 := x^2\left(\frac{\partial^2}{\partial x^2} y(x)\right) + x\left(\frac{\partial}{\partial x} y(x)\right) + \left(x^2 - \frac{1}{4}\right) y(x) = x^{(5/2)}\cos(x)$$

```
> K04:=duhamel(ode04,y,x,s);
```

$$K04 := \frac{\sqrt{s}\cos(s)\sin(x)}{\sqrt{x}} - \frac{\sqrt{s}\sin(s)\cos(x)}{\sqrt{x}}$$

```
> g04:=subs(x=s,rhs(ode04)/x^2);
```

$$g04 := \sqrt{s}\cos(s)$$

```
> int(K04*g04,s): y04:=simplify(subs(s=x,%));
```

$$y04 := \frac{1}{4}\sqrt{x}\,(\cos(x) + \sin(x)\,x)$$

Again let's compare our result with Maple's solution of ode04.

```
> dsolve(ode04,y(x));
```

$$y(x) = \frac{\_C1 \cos(x)}{\sqrt{x}} + \frac{\_C2 \sin(x)}{\sqrt{x}} + \frac{1}{4}\sqrt{x}\cos(x) + \frac{1}{4}x^{(3/2)}\sin(x)$$

Agreement again!